// Phong 光照模型 求高光反射
//phong  高光 = 高光颜色 * 高光系数 * pow(max(0,dot(反射向量,观察向量),高光范围))
// specular = specularColor * ks * pow(max(0,dot(R,V)),shininess)
Shader "Study/Chapter5/Phong"
{
    Properties
    {
        // 高光颜色
        _SpecularColor("SpecularColor",Color) = (1,1,1,1)
        // 高光系数
        _SpecularIntensity("SpecularIntensity",float) = 1
        // 高光范围
        _Shininess("Shininess",Range(0,10)) = 1
    }
    SubShader
    {
        Tags { "RenderType"="Opaque" }
        LOD 100

        Pass
        {
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
			#include "Lighting.cginc"
            #include "UnityCG.cginc"

            struct appdata
            {
                float4 vertex : POSITION;
				float3 normal : NORMAL;
            };

            struct v2f
            {
                float4 vertex : SV_POSITION;
				float3 worldPos : TEXCOORD0;
				float3 worldNormal : TEXCOORD1;
            };

			float4 _SpecularColor;
			float _SpecularIntensity;
			float _Shininess;


            v2f vert (appdata v)
            {
                v2f o;
                o.vertex = UnityObjectToClipPos(v.vertex);
				o.worldPos = mul(unity_ObjectToWorld,v.vertex);
				o.worldNormal = UnityObjectToWorldNormal(v.normal);
                return o;
            }

            fixed4 frag (v2f i) : SV_Target
            {

  				//phong  高光 = 高光颜色 * 高光系数 * pow(max(0,dot(反射向量,观察向量),高光范围))
                // specular = specularColor * ks * pow(max(0,dot(R,V)),shininess)

				float3 N = normalize(i.worldNormal);   // 法线归一化
				float3 L = normalize(_WorldSpaceLightPos0.xyz);			// 入射光线

				 // 求反射光
                // R = 2 * dot(N,L) * N - L;
				float3 R = reflect(-L,N);       // 反射向量
				float3 V = normalize(_WorldSpaceCameraPos.xyz - i.worldPos);   // 观察向量 顶点指向相机
				fixed RdotV = dot(R,V);

				fixed4 specular = _SpecularColor * _SpecularIntensity * pow(max(0,RdotV),_Shininess);
                return specular;
            }
            ENDCG
        }
    }
}
